MIT 14.32/14.320 **Spring 2023**
J. Angrist

Lecture Note 3

Confidence Intervals

1 Confidence Intervals for Means

With Normally distributed data,

$$T_n = \frac{\bar{X} - \mu_X}{s_X/\sqrt{n}} \sim t(n-1),$$

where s_X is the sample standard deviation of X_i. And statistic T_n (called a "t-statistic") is also asymptotically standard Normal regardless of the distribution of X_i.

Let c_α denote the critical value for a two-sided α-level test. For example, when $\alpha = .05$, $c_\alpha \approx 2$.

- A t-statistic is bracketed by $\pm c_n$ with probability $1 - \alpha$:

  $$P[-c_\alpha \leq T_n \leq c_\alpha] = 1 - \alpha$$

  Rearranging, we have:

  $$P[\bar{X} - c_\alpha(s_X/\sqrt{n}) \leq \mu_X \leq \bar{X} + c_\alpha(s_X/\sqrt{n})] = 1 - \alpha$$

- $s_X/\sqrt{n}$ is the *estimated standard error of the sample mean,* $\bar{X}$, in a sample of size n
- We say that

  $$[\bar{X} - c_\alpha(s_X/\sqrt{n}), \bar{X} + c_\alpha(s_X/\sqrt{n})]$$

  provides a $(1 - \alpha)$ confidence interval for μ_X.

- A 95% c.i. brackets μ_X 95% of the time. *Hey, what time is that?* When we draw repeated samples.
- But we don't draw repeated samples, we draw only one! Still, this probabilistic statement tells us how likely the interval is to *cover* the pop mean were we to conduct such a sampling experiment. Like standard errors, confidence intervals quantify *sampling variance.*
- Confidence intervals are random; Parameters (like μ_X) are fixed.
- In practice, we often ballpark the c.i.
  - A rule of thumb 95% confidence interval for a pop mean is given by the sample mean plus/minus two standard errors (since, if $\alpha = .05$, $c_\alpha = 1.96$ for Normal)

2 Confidence Intervals for Differences in Means

- The t-statistic for a *difference* in means is

$$T_n = \frac{(\bar{X}_1 - \bar{X}_2) - (\mu_1 - \mu_2)}{([s_1^2/n_1] + [s_2^2/n_2])^{1/2}} \sim t(n_1 + n_2 - 2),$$

where $T_n \sim N(0,1)$ for large n

- This comes from the fact that the *standard error of a difference in means* is

$$s^* = ([s_1^2/n_1] + [s_2^2/n_2])^{1/2}$$

  so,

$$P[(\bar{X}_1 - \bar{X}_2) - c_\alpha s^* \leq \mu_1 - \mu_2 \leq (\bar{X}_1 - \bar{X}_2) + c_\alpha s^*] = 1 - \alpha,$$

  where c_α is the $t(n_1 + n_2 - 2)$ critical value for a two-sided α-level test.

- A rule of thumb confidence interval for a difference in means is $(\bar{X}_1 - \bar{X}_2) \pm 2s^*$.

3 Confidence Intervals for Treatment Effects in ALO (2009)

TABLE 5—TREATMENT EFFECTS ON FIRST YEAR OUTCOMES IN THE SAMPLE WITH FALL GRADES

	SFP by type			Any SFP		
	All (1)	Men (2)	Women (3)	All (4)	Men (5)	Women (6)
Panel A. Fall grade						
Control mean	64.225	65.935	62.958	64.225	65.935	62.958
	(11.902)	(11.340)	(12.160)	(11.902)	(11.340)	(12.160)
SSP	0.349	−0.027	0.737	0.344	−0.014	0.738
	[0.917]	[1.334]	[1.275]	[0.917]	[1.332]	[1.274]
SFP	1.824	0.331	2.602			
	[0.847]**	[1.233]	[1.176]**			
SFSP	2.702	−0.573	4.205			
	[1.124]**	[2.010]	[1.325]***			
SFP (any)				2.125	0.016	3.141
				[0.731]***	[1.164]	[0.972]***
Observations	1,255	526	729	1,255	526	729
Panel B. First year GPA						
Control mean	1.805	1.908	1.728	1.797	1.885	1.731
	(0.902)	(0.908)	(0.891)	(0.904)	(0.910)	(0.894)
SSP	0.073	0.011	0.116	0.071	0.008	0.116
	[0.066]	[0.107]	[0.082]	[0.066]	[0.107]	[0.082]
SFP	0.010	−0.110	0.086			
	[0.064]	[0.103]	[0.084]			
SFSP	0.210	0.084	0.267			
	[0.092]**	[0.162]	[0.117]**			
SFP (any)				0.079	−0.042	0.147
				[0.056]	[0.095]	[0.073]**
Observations	1,255	526	729	1,255	526	729

Notes: The table reports regression estimates of treatment effects on full grades and first-year GPA computed using the full set of controls. Robust standard errors are reported in brackets. The sample is limited to students registered for at least two courses as of November 1 with data on the relevant set of controls and at least one fall grade. The last three columns report estimates from a model that combines the SFP and SFSP treatment groups into "SFP (any)."

* Significant at the 10 percent level.
** Significant at the 5 percent level.
*** Significant at the 1 percent level.

Ponder these questions, grasshopper:

- What determines c.i. width?
- Why are narrow intervals preferred to wide?
- Whats the connection between c.i. width and statistical power?

4 Monte Carlo Interval Coverage for Immigrant-Native Wage Diffs

- Asymptotic intervals cover well in real (definitely not Normal) data, with $n = 100$ only

```
------------------------------------------------------------------------------------------------------------------------------------
      name:  <unnamed>
       log:  /Users/joshangrist/Documents/teaching/14.32/SP2023/notes/LN3/LN3stata/LN3.log
  log type:  text
 opened on:  16 Feb 2023, 12:25:16

. 
.         /*
>         14.32 Lecture Note 3
>         c.i. coverage for immigrant-native wage gap
>         edited 02-16-23 to drop zeros and use unequal var ests
>         */
. 
.         // Insheet data from 2016 ACS PUMS
.         import delimited "ss16pusa.csv", clear
(encoding automatically selected: UTF-8)
(284 vars, 1,623,216 obs)

. 
.         // Sample of interest is men aged 40-49
.         keep if sex==1 & age>=40 & age <=49
(1,526,769 observations deleted)

. 
.         // Restrict sample to men who worked 50-52 weeks
.         keep if wkw==1
(24,305 observations deleted)

. 
.         // Generate usual hourly earnings
.         lab var wagp "Raw annual earnings"

.         lab var wkhp "Usual hours per week"

.         gen uhe = wagp/(50*wkhp)

.         quietly sum uhe, d

.         replace uhe=. if uhe>`r(p99)'
(819 real changes made, 819 to missing)

.         lab var uhe "Usual hourly earnings (truncated at 99pct)"

.         gen loguhe=log(uhe)
(4,963 missing values generated)

.         lab var loguhe "Log hourly earnings"

. 
.         // Code up immigrants
.         gen immig=(nativity==2)

.         lab var immig "Foreign born"

. 
. ttest loguhe, by(immig)

Two-sample t test with equal variances
------------------------------------------------------------------------------
   Group |     Obs        Mean    Std. err.   Std. dev.   [95% conf. interval]
---------+--------------------------------------------------------------------
       0 |  52,300    3.300842    .0031697     .724887    3.294629    3.307055
       1 |  14,879    3.137078    .0064108    .7819809    3.124512    3.149644
---------+--------------------------------------------------------------------
Combined |  67,179    3.264571     .002859    .7410341    3.258967    3.270175
---------+--------------------------------------------------------------------
    diff |            .1637641    .0068562                .1503259    .1772022
------------------------------------------------------------------------------
    diff = mean(0) - mean(1)                                      t =  23.8855
H0: diff = 0                                     Degrees of freedom =    67177

    Ha: diff < 0                 Ha: diff != 0                 Ha: diff > 0
 Pr(T < t) = 1.0000         Pr(|T| > |t|) = 0.0000          Pr(T > t) = 0.0000

. 
. gen popmu_native=r(mu_1)

. gen popmu_immig=r(mu_2)

. gen popdiff=popmu_native-popmu_immig

. 
.     // Keep only key variables in a newdata set called "workingextract"
. 
.         drop if missing(loguhe)
(4,963 observations deleted)

. 
.     keep wagp loguhe uhe wkhp immig age popmu_native popmu_immig popdiff

.         save workingextract, replace
file workingextract.dta saved

. 
. summarize

    Variable |        Obs        Mean    Std. dev.       Min        Max
-------------+---------------------------------------------------------
        agep |     67,179    44.59836    2.843473         40         49
        wagp |     67,179       77017    70468.34          4     665000
        wkhp |     67,179    44.73532    9.786132          1         99
         uhe |     67,179    33.90219    27.61486      .0016   201.5789
      loguhe |     67,179    3.264571    .7410341  -6.437752   5.306181
-------------+---------------------------------------------------------
       immig |     67,179    .2214829    .4152479          0          1
popmu_native |     67,179    3.300842           0   3.300842   3.300842
 popmu_immig |     67,179    3.137078           0   3.137078   3.137078
```

```
     popdiff |     67,179     .163764           0    .163764    .163764

. keep if _n==1
(67,178 observations deleted)

. keep popdiff

. save expresults, replace
file expresults.dta saved

. 
. /* draw samples of N=100 500 times compute means, diff, SEdiff */
. 
.    forvalues s=1/500 {
  2.         quietly use workingextract, clear
  3.         quietly bsample 100
  4.         quietly ttest loguhe, by(immig) unequal
  5.         quietly gen mu_nat=r(mu_1)
  6.     quietly gen mu_imm=r(mu_2)
  7.         quietly gen diff=mu_nat-mu_imm
  8.         quietly gen numNat=r(N_1)
  9.         quietly gen numImm=r(N_2)
 10.         quietly gen SEdiff=r(se)
 11.         quietly gen hi95 = diff + (SEdiff*1.96)
 12.         quietly gen lo95 = diff - (SEdiff*1.96)
 13.         quietly gen cover95=0
 14.         quietly replace cover95 = 1 if lo95<=popdiff &  popdiff<=hi95
 15.         quietly keep if _n==1
 16.         quietly append using expresults
 17.         quietly save expresults, replace
 18.    }

. 
.     keep popdiff mu_nat mu_imm diff numNat numImm SEdiff hi95 lo95 cover95

.     drop if missing(diff)
(1 observation deleted)

. 
.         /* sampling experiment results */
. 
.         list popdiff diff numNat numImm SEdiff lo95 hi95 cover95 if _n<=5

     +-------------------------------------------------------------------------------------+
     | popdiff        diff   numNat   numImm     SEdiff        lo95       hi95   cover95 |
     |-------------------------------------------------------------------------------------|
  1. | .163764    .0578763       76       24   .1725974   -.2804146   .3961673         1 |
  2. | .163764    .2617667       78       22   .1882998    -.107301   .6308343         1 |
  3. | .163764    .1345735       84       16   .1860601   -.2301043   .4992512         1 |
  4. | .163764    .4340758       79       21   .1863383    .0688528   .7992989         1 |
  5. | .163764   -.1437666       86       14   .1949807   -.5259289   .2383956         1 |
     +-------------------------------------------------------------------------------------+

.         summarize diff lo95 hi95 cover95

    Variable |        Obs        Mean    Std. dev.       Min        Max
-------------+---------------------------------------------------------
        diff |        500    .1713004    .1923067  -.3448374   .8456693
        lo95 |        500   -.1913058     .198277  -.7165971   .4191974
        hi95 |        500    .5339065    .2044576  -.0338668   1.797269
     cover95 |        500        .948    .2222494          0          1

. 
.         log close
      name:  <unnamed>
       log:  /Users/joshangrist/Documents/teaching/14.32/SP2023/notes/LN3/LN3stata/LN3.log
  log type:  text
 closed on:  16 Feb 2023, 12:26:34
-----------------------------------------------------------------------------------------------------------------------------------
```